

ERASE CHARACTER SET - SELECTED KEY IS C (E A)

```

0547 F8 LDI          ;Load RD (Utility
    48 07          ;Register) with
    49 BD PHI      RD ;Last address of
    4A F8 LDI          ;Character Set
    4B FF          ;      "      "
    4C AD PLO      RD ;      "      "
    4D ED SEX      RD ;X=RD
    4E F8 LDI          ;Load D with
    4F 00          ;00 - for erasing
0550 73 STXD        ;Store via X (RD) and decrement
    51 9D GHI      RD ;Get high byte of RD
    52 FB XRI          ;Exclusive OR it with
    53 05          ;05 to test if done
    54 3A BNZ        ;Branch if RD  $\neq$  05 (page above first
                        ;character set )
    55 4E          ;Branch to 054E - loop until done
    56 D4 SEP      R4 ;Return to CHIP-8 program

```

DEPOSIT SAMPLE CHARACTER IN MEMORY

```

0557 E2 SEX      R2 ;X=R2 (Stack pointer)
    58 22 DEC      R2 ;Point to free location
    59 F8 LDI          ;Load Utility Register RF
    5A 04          ;With loop count of 04
    5B AF PLO      RF ;And also put in RD (RD.1)
    5C BD PHI      RD ;Which will point to the storage
    5D F8 LDI          ;Area holding ASCII reference
    5E E6          ;Bit pattern address
    5F AD PLO      RD ;      "      "      "      - RD=04E6
0560 4D LDA      RD ;Get first part of address and advance pointer
    61 BC PHI      RC ;Store in RC.1
    62 OD LDN      RD ;Get second half of address
    63 AC PLO      RC ;Store in RC.0 (RC is indexed
                        ;to correct address)
    64 9B GHI      RB ;Display page
    65 BD PHI      RD ;Store in RD.1 (High order sample
                        ;character address)
    66 F8 LDI          ;Load Low order Sample
    67 93          ;Character address
    68 AD PLO      RD ;Into RD.0 (RD.0=0Y93)
    69 OD LDN      RD ;Get a row of the sample
    6A FE SHL          ;Shift left (for packing)
    6B FE SHL          ;      "      "      "      "
    6C FE SHL          ;      "      "      "      "
    6D FE SHL          ;      "      "      "      "
    6E 52 STR      R2 ;Push stack
    6F 8D GLO      RD ;Get RD.0 and
0570 FC ADI          ;Add 08 for next
    71 08          ;row
    72 AD PLO      RD ;Put adjusted value back in RD.0

```